Sarjaratkaisun etsiminen Mathematicalla

Olkoon tarkasteltavana ensimmäisen kertaluvun differentiaaliyhtälö:

```
diffyht = y'[x] == 1 + y[x]^2
```

$$y'[x] == 1 + y[x]^2$$

Tälle pyritään etsimään sarjaratkaisu origokeskisenä potenssisarjana. Tavoitteena on laskea sarjan termien kertoimet *n*-asteiseen termiin saakka:

```
n = 10
```

10

Tarvittava potenssisarjamuotoinen yrite jäännöstermeineen on

```
yrite := Function[x, Sum[a_k x^k, {k, 0, n}] + O[x]^(n + 1)]
```

Tämä on esitetty funktiona eikä lausekkeena, jotta yhtälöön sijoitettaessa myös derivaatta saadaan sijoitetuksi:

```
sarjayht = diffyht /. y -> yrite
```

Series::sbyc : Division by a series with no coefficients in $\frac{1}{O[x]^1}$. More…

$$a_1 + 2 a_2 x + 3 a_3 x^2 + 4 a_4 x^3 + 5 a_5 x^4 + 6 a_6 x^5 + 7 a_7 x^6 + 8 a_8 x^7 + 9 a_9 x^8 + 10 a_{10} x^9 + O[x]^{10} == (1 + a_0^2) + 2 a_0 a_1 x + (a_1^2 + 2 a_0 a_2) x^2 + (2 a_1 a_2 + 2 a_0 a_3) x^3 + (a_2^2 + 2 a_1 a_3 + 2 a_0 a_4) x^4 + (2 a_2 a_3 + 2 a_1 a_4 + 2 a_0 a_5) x^5 + (a_3^2 + 2 a_2 a_4 + 2 a_1 a_5 + 2 a_0 a_6) x^6 + (2 a_3 a_4 + 2 a_2 a_5 + 2 a_1 a_6 + 2 a_0 a_7) x^7 + (a_4^2 + 2 a_3 a_5 + 2 a_2 a_6 + 2 a_1 a_7 + 2 a_0 a_8) x^8 + (2 a_4 a_5 + 2 a_3 a_6 + 2 a_2 a_7 + 2 a_1 a_8 + 2 a_0 a_9) x^9 + (a_5^2 + 2 a_4 a_6 + 2 a_3 a_7 + 2 a_2 a_8 + 2 a_1 a_9 + 2 a_0 a_{10}) x^{10} + O[x]^{11}$$

Yhtälön oikealla ja vasemmalla puolella on potenssisarja. Jotta nämä olisivat samat, tulee samakorkuisten potenssien kertoimien yhtälön oikealla ja vasemmalla puolella olla samat. Kertoimien välille tällöin syntyvät yhtälöt voidaan muodostaa Mathematican `LogicalExpand`-komennolla. Ensimmäinen yhtälö saadaan vertaamalla vakiotermien kertoimia, toinen vertaamalla ensimmäisen asteen termejä jne.

```
yhtalot = LogicalExpand[sarjayht]
```

$$1 + a_0^2 - a_1 == 0 \;\&\&\; 2 a_0 a_1 - 2 a_2 == 0 \;\&\&\; a_1^2 + 2 a_0 a_2 - 3 a_3 == 0 \;\&\&\; 2 a_1 a_2 + 2 a_0 a_3 - 4 a_4 == 0 \;\&\&\; a_2^2 + 2 a_1 a_3 + 2 a_0 a_4 - 5 a_5 == 0 \;\&\&\; 2 a_2 a_3 + 2 a_1 a_4 + 2 a_0 a_5 - 6 a_6 == 0 \;\&\&\; a_3^2 + 2 a_2 a_4 + 2 a_1 a_5 + 2 a_0 a_6 - 7 a_7 == 0 \;\&\&\; 2 a_3 a_4 + 2 a_2 a_5 + 2 a_1 a_6 + 2 a_0 a_7 - 8 a_8 == 0 \;\&\&\; a_4^2 + 2 a_3 a_5 + 2 a_2 a_6 + 2 a_1 a_7 + 2 a_0 a_8 - 9 a_9 == 0 \;\&\&\; 2 a_4 a_5 + 2 a_3 a_6 + 2 a_2 a_7 + 2 a_1 a_8 + 2 a_0 a_9 - 10 a_{10} == 0$$

Tuloksena on rekursiivinen epälineaarinen yhtälöryhmä: ensimmäisestä yhtälöstä voidaan ratkaista a_1, jos a_0 tunnetaan, toisesta tämän jälkeen a_2, kolmannesta a_3 jne. Kokonaisuudesaan ryhmä voidaan ratkaista `Solve`-komennolla. Tässä jälkimmäiseksi argumentiksi pitäisi oikeastaan antaa lista yhtälöryhmän tuntemattomista, mutta oletuksena on, että ratkaisu tapahtuu kaikkien ryhmässä esiintyvien symbolien suhteen. Kaikkia tuntemattomia ei saada ratkaistuiksi, vaan muut voidaan ainoastaan lausua ensimmäisen kertoimen a_0 avulla; tästä annetaan varoitus.

```
ratk = Solve[yhtalot]
```

Solve::svars : Equations may not give solutions for all "solve" variables. More…

$$\left\{\left\{a_{10} \to \frac{a_0\,(1382 + 14102\,a_0^2 + 47685\,a_0^4 + 72765\,a_0^6 + 51975\,a_0^8 + 14175\,a_0^{10})}{14175},\right.\right.$$
$$a_9 \to \frac{62 + 1382\,a_0^2 + 6360\,a_0^4 + 11655\,a_0^6 + 9450\,a_0^8 + 2835\,a_0^{10}}{2835},$$
$$a_8 \to \frac{1}{315}\,a_0\,(62 + 440\,a_0^2 + 1008\,a_0^4 + 945\,a_0^6 + 315\,a_0^8),$$
$$a_7 \to \frac{1}{315}\,(17 + 248\,a_0^2 + 756\,a_0^4 + 840\,a_0^6 + 315\,a_0^8),$$
$$a_6 \to \frac{1}{45}\,a_0\,(17 + 77\,a_0^2 + 105\,a_0^4 + 45\,a_0^6),\ a_5 \to \frac{1}{15}\,(2 + 17\,a_0^2 + 30\,a_0^4 + 15\,a_0^6),$$
$$\left.\left.a_4 \to \frac{1}{3}\,a_0\,(2 + 5\,a_0^2 + 3\,a_0^4),\ a_3 \to \frac{1}{3}\,(1 + 4\,a_0^2 + 3\,a_0^4),\ a_2 \to a_0\,(1 + a_0^2),\ a_1 \to 1 + a_0^2\right\}\right\}$$

Sijoittamalla kertoimet yritteeseen saadaan ratkaisuna olevan potenssisarjan alkupää:

```
sarjaratk = yrite[x] /. First[ratk]
```

$$a_0 + (1 + a_0^2)\,x + a_0\,(1 + a_0^2)\,x^2 + \frac{1}{3}\,(1 + 4\,a_0^2 + 3\,a_0^4)\,x^3 + \frac{1}{3}\,a_0\,(2 + 5\,a_0^2 + 3\,a_0^4)\,x^4 +$$
$$\frac{1}{15}\,(2 + 17\,a_0^2 + 30\,a_0^4 + 15\,a_0^6)\,x^5 + \frac{1}{45}\,a_0\,(17 + 77\,a_0^2 + 105\,a_0^4 + 45\,a_0^6)\,x^6 +$$
$$\frac{1}{315}\,(17 + 248\,a_0^2 + 756\,a_0^4 + 840\,a_0^6 + 315\,a_0^8)\,x^7 + \frac{1}{315}\,a_0\,(62 + 440\,a_0^2 + 1008\,a_0^4 + 945\,a_0^6 + 315\,a_0^8)\,x^8 +$$
$$\frac{(62 + 1382\,a_0^2 + 6360\,a_0^4 + 11655\,a_0^6 + 9450\,a_0^8 + 2835\,a_0^{10})\,x^9}{2835} +$$
$$\frac{a_0\,(1382 + 14102\,a_0^2 + 47685\,a_0^4 + 72765\,a_0^6 + 51975\,a_0^8 + 14175\,a_0^{10})\,x^{10}}{14175} + O[x]^{11}$$

Tämä sisältää yhden määräämättömän vakion a_0, kuten ensimmäisen kertaluvun differentiaaliyhtälön yleiselle ratkaisulle luonnollista onkin. Jos alkuehdoksi valitaan $y(0) = 0$, tulee olla $a_0 = 0$. Vastaava yksittäisratkaisu on

```
yksittratk = sarjaratk /. a0 -> 0
```

$$x + \frac{x^3}{3} + \frac{2\,x^5}{15} + \frac{17\,x^7}{315} + \frac{62\,x^9}{2835} + O[x]^{11}$$

Tämä on sama kuin funktion tan x origokeskinen Taylorin sarja:

```
Series[Tan[x], {x, 0, n}]
```

$$x + \frac{x^3}{3} + \frac{2\,x^5}{15} + \frac{17\,x^7}{315} + \frac{62\,x^9}{2835} + O[x]^{11}$$

Näin tulee ollakin, sillä differentiaaliyhtälö on separoituva ja sen yleiseksi ratkaisuksi saadaan $y = \tan(x + C)$:

```
DSolve[diffyht, y[x], x]
```

{{y[x] → Tan[x + C[1]]}}

Jotta sarjaratkaisulle voitaisiin piirtää kuvaaja, siitä on pudotettava jäännöstermi pois. Tämän tarkkaa lausekettahan ei tunneta eikä sille siis voida laskea numeerisia arvoja piirtämistä varten. Jäännöstermin poistaminen tapahtuu komennolla `Normal`:

```
poly = Normal[yksittratk]
```

$$x + \frac{x^3}{3} + \frac{2\,x^5}{15} + \frac{17\,x^7}{315} + \frac{62\,x^9}{2835}$$

Sarjaratkaisu ja funktio tan *x* samassa kuvassa:

```
Plot[{poly, Tan[x]}, {x, 0, 1.5}]
```



- Graphics -

Esitetty lasku ei anna viitteitä sarjaratkaisun suppenemisalueesta. Kokonaan muilla keinoilla voidaan osoittaa, että tangentin origokeskisen Taylorin sarjan suppenemissäde on $\pi/2$. Sarja suppenee siis vain välillä $]-\pi/2,\ \pi/2[$.

Haluttua termilukua *n* voidaan edellä olevassa laskussa muuttaa ja tämän jälkeen laskea kaikki uudelleen valinnalla *Evaluate Notebook* valikosta *Kernel/Evaluation*.

Linkit

sarjamuotoinen yrite

SKK 30.04.2001